


Apple II® BASIC

常用程式設計

王榮輝・陳文堅 合譯



協群科技出版社

Apple II® BASIC

常用程式設計

王 榮 輝 合譯
陳 文 堅

協 群 科 技 出 版 社

Apple II® BASIC

常用程式設計

編譯者：王榮輝·陳文堅

出版：協群科技出版社

發行：協群科技出版社

香港中環卑利街684號二樓

印刷者：廣源印務局

青山道875號工廠大廈

定價：H.K.\$ 18.00

簡 介

本書收集了 76 個實際應用程式（由 Applesoft BASIC 所寫的），使用這些程式可以處理多種普遍且實際的工作。

因為目前已經有許多書提到關於如何使用 BASIC 來設計程式，所以在本書當中，並不企圖教導學習者如何用 BASIC 來設計程式。但卻詳細說明本書所採用的程式，包括例題及將原始程式列印出來。

讀者不論是否熟悉 BASIC 這種程式語言，均可以使用這本書，若不熟悉的話，只要將所列的程式輸入電腦內，然後按照書上舉的例子來執行，即可瞭解程式的功用為何。

在每個程式中均有詳細的註解來幫助程式設計者了解每個程式的邏輯，並且程式設計者可以在設計自己的程式時，應用這些程式中的一些技巧。在程式當中，所有的註解式子均在所說明之程式部份前面。而在輸入程式時，這些註解式子均可省略，因為註解式子對電腦而言不僅浪費記憶位置且是無意義的。

在某些程式中尚包括有選擇部份（OPTION），這些選擇部份主要是修改原來程式的輸入或輸出格式。對於這些選擇部份，本書均附有概要說明，例子，並將修改部份的程式列印出來，包括要增加或刪除的式子在內。

所有的程式在 Apple II 或 Apple II Plus 的機器上均可執行，只要這些機器上有 Applesoft BASIC 的編譯程式。

在本書中，幾乎所有的程式均是將結果輸出於終端機上，除了少數幾個程式是將結果輸出於報表上。同時將結果輸出於報表上的程式，均假設其列表機的介面是插在插座 1 上，且輸出指令是使用 DOS 的控制指令來控制。若所執行的程式不使用 DOS 指令，則輸出指令形式要改變成較簡單形式，例如 PRINT D\$；“PR#1”改為 PR#1，則輸出結果將直接經由插座 1 輸出。

所有程式均可修改，然後將程式及結果自報表紙輸出，但要注意，程式中的 VTAB，HTAB（用來控制亮點的移動），必須除去。且 TAB 指令，對大多數列表機來講是無效的，應改為 POKE 36,t 的形式，其中 t = 要空的格數。

將結果自報表上輸出後，要將結果再從終端機上輸出，則使用 PR#Q 指令即可。

在本書中，所有 FOR-NEXT 迴路指令形式，均如下所示：

```
FOR A = 1 TO N
NEXT A
```

現在 Applesoft BASIC 中 NEXT 後面的變數是可以省略的，將此變數寫出來的用意是讓讀者比較容易看懂程式。

在 Applesoft BASIC 中，當輸入 "PRINT" 時可用 "?" 來代替，以提高輸入速度。

在某些程式中，有錯誤檢查式子，當遇到錯誤時，鈴會響一下，這是用 PRINT CHR\$(7) 這個指令來達到使鈴響的目的。在程式當中亦可鍵入 CTRL-G 於兩個雙引號間，來控制使鈴響。

所有於本書中的程式均被測試並執行過（在 Apple II 機器，使用 Applesoft BASIC 編譯程式）

目 錄

投資(存款)的期望值	1	函數繪圖	79
定期存款的期望值	3	線性內插法	83
期望固定回收額下的存款	5	曲線內插法	85
存款中固定的提款	7	積分：辛浦森法則	87
固定期望值下的最初投資額	9	積分：梯形法則	90
固定提款下之存款額	11	積分：高斯永積法	92
投資的利率	13	導函數	94
投資中的有效利率	15	求二次方程式之實根	95
利息收入表	17	多項式的實根：牛頓法	97
折舊率	22	多項式之實根：區間減半法	99
貶值總額	23	三角函數多項式	101
殘值的計算	25	聯立方程式	103
支票兌現額的計算	27	線性規劃	105
貸款的金額	28	矩陣之加法、減法和係數乘積	110
貸款中不變的支付	30	矩陣乘法	113
貸款的最後支付額	32	反矩陣	116
貸款中的餘額	35	排列組合	119
貸款中償還的年數	38	Mann-Whitney U 測試	121
貸款中的利率	40	平均數、變異數、標準差	124
抵押攤還表	43	等比中項和標準差	127
最大公因數	49	二項分佈	129
整數之質因數	51	卜瓦松分佈	131
多邊形面積	53	常態分佈	132
三角形	55	卡方分佈	134
兩個向量的分析	60	卡方測試	137
二個向量間之運算	62	學生氏 t —— 分佈	139
角度之轉換：徑度量轉換成角度量	63	學生氏 t —— 分佈測試	141
角度之轉換：角度量轉換成徑度量	65	F —— 分佈	144
直角座標——極座標的轉換	67	線性相關係數	147
座標繪圖	69	線性迴歸	149
極座標繪圖	74	多重線性迴歸	151

多項式迴歸.....	155
幾何迴歸.....	158
指數迴歸.....	161
系統信賴度.....	163
平均成長率.....	165
折舊金額表.....	167
書寫發票.....	171
處方成本之計算.....	175
圖形檢查.....	179
求某一日期為星期幾.....	186
兩日期間之天數.....	188
英制、公制之轉換.....	191
字母排列.....	194

投資（存款）的期望值

本程式用來計算一項投資（存款）的可能價值，在執行此程式時必須要有利率（ i ），期初投資額（ P ），每年複利計算次數（ N ），投資的年數（ Y ）等資料。

假設沒有額外的資本主投資或提取之情況下，這未來可能價值將可由以下公式算出：

$$F = P(1 + i/N)^{N \cdot Y}$$

其中 T ： Y 年後的總值

P ：期初投資額

i ：利率

N ：每年複利計算次數

Y ：年數

例題

王先生存入 6800 元於銀行帳戶，假設利率為 9.5%，如果每年四次複利計算，十年後他的存款額將有多少？

張小姐購買某公司的股票 16050 元，這股票的價值依每年 7% 的平均增加率上漲，在 5 年半後張小姐這股票將值多少？

程式執行後的結果：

FUTURE VALUE OF AN INVESTMENT

```
INITIAL INVESTMENT :6800
NOMINAL INTEREST RATE :9.5
# OF COMPOUNDING PERIODS PER YEAR :4
NUMBER OF YEARS :10
FUTURE VALUE = $17388.64
```

MORE DATA (Y=YES, N=NO): Y

```
INITIAL INVESTMENT :16050
NOMINAL INTEREST RATE :7
# OF COMPOUNDING PERIODS PER YEAR :1
NUMBER OF YEARS :5.5
FUTURE VALUE = $23285.51
```

MORE DATA (Y=YES, N=NO): N

程式報表：

```
10 HOME : PRINT TAB( 5)"FUTURE VALUE OF AN INVESTMENT"
20 PRINT
29 REM STATEMENTS 30 TO 90 REQUEST USER INPUT
30 INPUT "INITIAL INVESTMENT :";P
```

```

50 INPUT "NOMINAL INTEREST RATE :";I
80 INPUT "# OF COMPOUNDING PERIODS PER YEAR :";N
90 INPUT "NUMBER OF YEARS :";Y
108 REM CALCULATE INTEREST RATE PER PERIOD
109 REM CONVERT FROM PERCENT TO DECIMAL
110 I = I / N / 100
119 REM CALCULATE FUTURE VALUE BY FORMULA
120 T = P * (1 + I) ^ (N * Y)
129 REM ROUND OFF TO NEAREST CENT, PRINT
130 PRINT "FUTURE VALUE = $"; INT (T * 100 + .5) / 100
140 PRINT
149 REM RESTART OR END PROGRAM ? USER INPUT REQUIRED
150 INPUT "MORE DATA (Y=YES, N=NO): ";Z$: IF Z$ = "Y" THEN 20
160 END

```

問題的變型

上述都以每年為一個計算單位，這節中將討論到以整年度或幾個月的計算方式，有關於公式部份的改變請參照下例。

例題

王先生以 8 % 的利率投資 12000 元，每年四次複利計算，在 10 年又 7 個月之後他這項投資的總值為多少？

程式執行後的結果：

FUTURE VALUE OF AN INVESTMENT

```

INITIAL INVESTMENT :12000
NOMINAL INTEREST RATE :8
# OF COMPOUNDING PERIODS PER YEAR :4
NUMBER OF YEARS :10
NUMBER OF MONTHS :7
FUTURE VALUE = $27749.5

```

MORE DATA (Y=YES, N=NO): N

程式報表：

```

1 REM OPTION 90-105
10 HOME : PRINT TAB( 5)"FUTURE VALUE OF AN INVESTMENT"
:
80 INPUT "# OF COMPOUNDING PERIODS PER YEAR :";N
90 INPUT "NUMBER OF YEARS :";Y0
100 INPUT "NUMBER OF MONTHS :";M
104 REM CALCULATE YEARS FROM YEARS AND MONTHS
105 Y = (12 * Y0 + M) / 12
108 REM CALCULATE INTEREST RATE PER PERIOD
:
160 END

```

定期存款的期望值

此程式是計算定期存入而且每一次數額相等的未來期望值，使用本程式時需代入固定存款額（R），每年存款次數（N），利率（i），年數（Y）等已知條件。

假設每次存入依複利計算公式如下：

$$T = R \times \left(\frac{(1 + i / N)^{N \times Y} - 1}{i / N} \right)$$

其中 T：Y年後總值

R：固定存款額

N：每年存款次數

Y：年數

i：利率

例題

張先生每月由活期存款中撥出 50 元存入台灣銀行的定期存款帳戶，利率 5% 年底張先生將可收到多少錢？

李先生每年存 175 元到華南銀行的定期存款儲蓄帳戶，利率 5.5%，15 年後累積多少？

程式執行後的結果：

FUTURE VALUE OF REGULAR DEPOSITS (ANNUITY)

```
AMOUNT OF REGULAR DEPOSITS :50
NOMINAL INTEREST RATE :5
NUMBER OF DEPOSITS PER YEAR :12
NUMBER OF YEARS :1
FUTURE VALUE = $613.94
```

MORE DATA ? (Y=YES N=NO):Y

```
AMOUNT OF REGULAR DEPOSITS :175
NOMINAL INTEREST RATE :5.5
NUMBER OF DEPOSITS PER YEAR :1
NUMBER OF YEARS :15
FUTURE VALUE = $3921.52
```

MORE DATA ? (Y=YES N=NO):N

程式報表：

```
10 HOME : PRINT TAB( 4 )"FUTURE VALUE OF REGULAR DEPOSITS": PRINT
    TAB( 15 )" (ANNUITY) "
20 PRINT
29 REM STATEMENTS 30 TO 90 REQUEST USER INPUT
30 INPUT "AMOUNT OF REGULAR DEPOSITS :";R
50 INPUT "NOMINAL INTEREST RATE :";I
70 INPUT "NUMBER OF DEPOSITS PER YEAR :";N
90 INPUT "NUMBER OF YEARS :";Y
```

```

108 REM CALCULATE INTEREST RATE PER DEPOSIT
109 REM CONVERT FROM PERCENT TO DECIMAL
110 I = I / N / 100
111 REM CALCULATE FUTURE VALUE BY FORMULA
120 T = R * ((1 + I) ^ (N * Y) - 1) / I
129 REM ROUND OFF TO NEAREST CENT, PRINT
130 PRINT "FUTURE VALUE = $"; INT (T * 100 + .5) / 100
139 REM PRINT BLANK LINE TO SEPARATE DATA FROM QUESTION
140 PRINT
150 INPUT "MORE DATA ? (Y=YES N=NO):"; Z%
170 IF Z% = "Y" THEN 20
180 END

```

問題的變型

如上節所述，若欲自由選擇幾年幾個月為計算單位，而不是以年度為計算單位者，程式部份的改變可由下例看出。

例題

王先生每個月過戶 50 元到信用投資公司，10 年又 5 個月後，依 5 % 利率，他將會收到多少錢？

程式執行後的結果：

FUTURE VALUE OF REGULAR DEPOSITS
(ANNUITY)

```

      AMOUNT OF REGULAR DEPOSITS :50
      NOMINAL INTEREST RATE :5
      NUMBER OF DEPOSITS PER YEAR :12
      NUMBER OF YEARS :10
      NUMBER OF MONTHS :5
      FUTURE VALUE = $8179.31

```

MORE DATA ? (Y=YES N=NO):N

程式報表：

```

1 REM OPTION 90-105
10 HOME : PRINT TAB( 4) "FUTURE VALUE OF REGULAR DEPOSITS": PRINT
   TAB( 15) "(ANNUITY)"
:
70 INPUT "NUMBER OF DEPOSITS PER YEAR:";N
90 INPUT "NUMBER OF YEARS:";Y0
100 INPUT "NUMBER OF MONTHS:";M
104 REM CALCULATE YEARS FROM YEARS AND MONTHS
105 Y = (12 * Y0 + M) / 12
108 REM CALCULATE INTEREST RATE PER DEPOSIT
:
180 END

```

期望固定回收額下的存款

此程式是計算在一定時間內，期望固定回收額的情況下，定期存款的數額，當然每年存款次數（N），期望回收額（T），利率（i），及年數（Y）等資料是必需的。

由前面的公式可導出：

$$R = T \left(\frac{i/N}{(1 + i/N)^{N \cdot Y} - 1} \right)$$

其中，R：固定存款的數額

T：期望回收數額

i：利率

N：每年存款次數

Y：年數

例題

張小姐希望在一年後她的存款上有1000元，她每月要存多少方能到達呢？（在利率8%之下）

程式執行後的結果：

REGULAR DEPOSITS

TOTAL VALUE AFTER 'Y' YEARS :1000

NOMINAL INTEREST RATE :8

NUMBER OF DEPOSITS PER YEAR :12

NUMBER OF YEARS :1

REGULAR DEPOSITS = \$80.32

MORE DATA ? (Y=YES N=NO):N

程式報表：

```

10 HOME : PRINT TAB( 12)"REGULAR DEPOSITS"
20 PRINT
29 REM STATEMENTS 30 TO 90 REQUEST USER INPUT
30 INPUT "TOTAL VALUE AFTER 'Y' YEARS :";T
50 INPUT "NOMINAL INTEREST RATE :";I
70 INPUT "NUMBER OF DEPOSITS PER YEAR :";N
90 INPUT "NUMBER OF YEARS :";Y
108 REM CALCULATE INTEREST RATE PER DEPOSIT
109 REM CONVERT FROM PERCENT TO DECIMAL
110 I = I / N / 100
119 REM CALCULATE AMOUNT OF REGULAR DEPOSIT BY FORMULA
120 R = T * I / ((1 + I) ^ (N * Y) - 1)
129 REM ROUND OFF TO NEAREST CENT. PRINT

```

```

130 PRINT "REGULAR DEPOSITS = $"; INT (R * 100 + .5) / 100
140 PRINT
150 INPUT "MORE DATA ? (Y=YES N=NO):";Z$
170 IF Z$ = "Y" THEN 20
180 END

```

問題的變型

如果期望回收的期間不是恰好以年為單位，則可將程式稍作修改：

例題

朱先生想要在一年又五個月中存 2000 元去買一部機車，假如銀行的利率是 8 % 那他每月應存多少呢？

程式執行後的結果：

```

          REGULAR DEPOSITS

TOTAL VALUE AFTER 'Y' YEARS :2000
NOMINAL INTEREST RATE :8
NUMBER OF DEPOSITS PER YEAR :12
NUMBER OF YEARS :1
NUMBER OF MONTHS :5
REGULAR DEPOSITS = $111.5

MORE DATA ? (Y=YES N=NO):N

```

程式報表：

```

1  REM      OPTION 90-105
10 HOME : PRINT TAB( 12)"REGULAR DEPOSITS"
:
70 INPUT "NUMBER OF DEPOSITS PER YEAR:";I
90 INPUT "NUMBER OF YEARS :";Y0
100 INPUT "NUMBER OF MONTHS :";M
104 REM      CALCULATE YEARS FROM YEARS AND MONTHS
105 Y = (12 * Y0 + M) / 12
108 REM      CALCULATE INTEREST RATE PER DEPOSIT
:
180 END

```

存款中固定的提款

此程式是計算在一定期間內固定地從存款中提款的最大數額，假設所有提款數都相等。我們必需供應最初的存款額（ P ），利率（ i ），每年的提款次數（ N ）和年數（ Y ）等。

最大提款數額之計算是根據下列公式：

$$R = P \left(\frac{i/N}{(1+i/N)^{N \cdot Y}} - \frac{i}{1+i/N} \right)$$

其中， R ：固定提款數額

P ：最初存款額

i ：利率

N ：每年提款次數

Y ：年數

因為這程式是計算每月最大的提款數，所以到最後存款將一無所有，你可以提出比此數目較少的數額而在最後仍留有存款。

例題

王先生在利率 9.5 % 下存入 8000 元，他打算在十年中每個月提出固定的款項，而且完全提完，他每次應提多少呢？

程式執行後的結果：

REGULAR WITHDRAWALS FROM AN INVESTMENT

INITIAL INVESTMENT :8000

NOMINAL INTEREST RATE :9.5

NUMBER OF WITHDRAWALS PER YEAR :12

NUMBER OF YEARS :10

AMOUNT OF EACH WITHDRAWAL = \$103.52

MORE DATA ? (Y=YES N=NO):N

程式報表：

```
10 HOME : PRINT " REGULAR WITHDRAWALS FROM AN INVESTMENT"
20 PRINT
25 REM STATEMENTS 30 TO 90 REQUEST USER INPUT
30 INPUT "INITIAL INVESTMENT :";P
50 INPUT "NOMINAL INTEREST RATE :";I
70 INPUT "NUMBER OF WITHDRAWALS PER YEAR :";N
90 INPUT "NUMBER OF YEARS :";Y
```

```

108 REM CALCULATE INTEREST RATE PER WITHDRAWAL
109 REM CONVERT FROM PERCENT TO DECIMAL
110 I = I / N / 100
119 REM CALCULATE REGULAR WITHDRAWAL BY FORMULA
120 R = P * (I / ((1 + I) ^ (N * Y) - 1) + I)
129 REM ROUND OFF TO NEAREST CENT, PRINT
130 PRINT "AMOUNT OF EACH WITHDRAWAL = $"; INT (R * 100 + .5) / 100
140 PRINT
149 REM RESTART OR END PROGRAM ?
150 INPUT "MORE DATA ? (Y=YES N=NO):";Z$
170 IF Z$ = "Y" THEN 20
180 END

```

問題的變型

如果我們希望每次提款的时间是以年與月為單位，而不僅僅是以年為單位，則可將程式做以下的修改：

例題

假如你有 8000 元，以 9.5 % 的利率投資，分十年又五個月的期間來提款，你每個星期可提多少？

程式執行後的結果：

REGULAR WITHDRAWALS FROM AN INVESTMENT

```

INITIAL INVESTMENT :8000
NOMINAL INTEREST RATE :9.5
NUMBER OF WITHDRAWALS PER YEAR :52
NUMBER OF YEARS :10
NUMBER OF MONTHS :5
AMOUNT OF EACH WITHDRAWAL = $23.28

```

MORE DATA ? (Y=YES N=NO):N

程式報表：

```

1 REM OPTION 90-105
10 HOME : PRINT " REGULAR WITHDRAWALS FROM AN INVESTMENT"
11 :
70 "NUMBER OF WITHDRAWALS PER YEAR :";N
90 INPUT "NUMBER OF YEARS :";Y0
100 INPUT "NUMBER OF MONTHS :";M
104 REM CALCULATE YEARS FROM YEARS AND MONTHS
105 Y = (12 * Y0 + M) / 12
108 REM CALCULATE INTEREST RATE PER WITHDRAWAL
109 :
180 END

```

固定期望值下的最初投資額

此程式是計算在某一段期間後，預期某一回收額之下的最初投資額。當然公式中的期望值（ T ），每年複利計算次數（ N ），年數（ Y ），利率（ i ）等資必需為已知數。
計算最初投資額公式如下：

$$P = \frac{T}{(1 + i/N)^{N \cdot Y}}$$

其中： P ：最初投資額
 T ：期望值
 N ：每年複利計算次數
 Y ：年數
 i ：利率

例題

如果是以每季來計算複利，要在十年後（在利率 8.5 % 下）達到 10000 元，最初必須投資多少？

張先生希望五年後能有 5000 元，假設每天以複利計算利率為 7.9 %，則他必須存入多少？

程式執行後的結果：

INITIAL INVESTMENT

```
TOTAL VALUE AFTER 'Y' YEARS :10000
# OF COMPOUNDING PERIODS PER YEAR :4
# OF YEARS :10
NOMINAL INTEREST RATE :8.5
INITIAL INVESTMENT = $4312.38
```

MORE DATA ? (Y=YES N=NO):Y

```
TOTAL VALUE AFTER 'Y' YEARS :5000
# OF COMPOUNDING PERIODS PER YEAR :365
# OF YEARS :5
NOMINAL INTEREST RATE :7.9
INITIAL INVESTMENT = $3368.54
```

MORE DATA ? (Y=YES N=NO):N

程式報表：

```
10 HOME : PRINT TAB( 11)"INITIAL INVESTMENT"
20 PRINT
29 REM STATEMENTS 30 TO 90 REQUEST USER INPUT
30 INPUT "TOTAL VALUE AFTER 'Y' YEARS :";T
50 INPUT "# OF COMPOUNDING PERIODS PER YEAR :";N
70 INPUT "# OF YEARS :";Y
90 INPUT "NOMINAL INTEREST RATE :";I
108 REM CALCULATE INTEREST RATE PER PERIOD
```

```

109 REM CONVERT FROM PERCENT TO DECIMAL
110 I = I / N / 100
119 REM CALCULATE INITIAL INVESTMENT BY FORMULA
120 P = T / (1 + I) ^ (N * Y)
129 REM ROUND OFF TO NEAREST CENT, PRINT
130 PRINT "INITIAL INVESTMENT = $"; INT (P * 100 + .5) / 100
140 PRINT
149 REM RESTART OR END PROGRAM?
150 INPUT "MORE DATA ? (Y=YES N=NO):"; Z$
170 IF Z$ = "Y" THEN 20
180 END

```

問題的變型

上述程式乃是以整年為一個計算單位，也許在實用上並不一定都以整年為計算單位，（如下例）則可將程式作適當的修改：

例題

趙先生想在銀行開戶存款，他希望在三年又八個月後，他的戶頭中有4000元，假設每月以8%複利計算，則他必須存入多少？

程式執行後的結果：

```

          INITIAL INVESTMENT

TOTAL VALUE AFTER 'Y' YEARS :4000
# OF COMPOUNDING PERIODS PER YEAR :12
NUMBER OF YEARS :3
NUMBER OF MONTHS :8
NOMINAL INTEREST RATE :8
INITIAL INVESTMENT = $2986

MORE DATA ? (Y=YES N=NO):N

```

程式報表：

```

1 REM OPTION 70-85
10 HOME : PRINT TAB(11)"INITIAL INVESTMENT"
:
50 INPUT "# OF COMPOUNDING PERIODS PER YEAR :";Y0
70 INPUT "NUMBER OF YEARS :";Y
80 INPUT "NUMBER OF MONTHS :";M
84 REM CALCULATE YEARS FROM YEARS AND MONTHS
85 Y = (12 * Y0 + M) / 12
90 INPUT "NOMINAL INTEREST RATE :";I
108 REM CALCULATE INTEREST RATE PER PERIOD
:
180 END

```